// File: contracts/ConspiracyDAO.sol



// ********************************************* //
// ********       Our Social Handles     ****** //
// ******************************************* //




// https://twitter.com/TokenConspiracy

// https://t.me/ConspiracyDAO

// Conspiracydaotoken.com




// ********************************************* //


// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;
import "./ConspiracyDAOToken.sol";

contract ConspiracyDAO is ConspiracyDAOToken {
    struct Conspiracy{
        uint256 id;
        address creator;
        string name;
        string description;
        string imageUri;
        uint256 upvotes;
        uint256 downvotes;
    }

    uint256 public createConspiracyCost = 100e18;

    Conspiracy[] private conspiracies;

    // voterAddress => conspiracy Id => upvoted bool
    mapping(address => mapping(uint256 => bool)) private upvoted;
    // voterAddress => conspiracy Id => downvoted bool
    mapping(address => mapping(uint256 => bool)) private downvoted;

    //////////////////////////////////////////
    //////////     Events     ///////////////
    ////////////////////////////////////////
    event CreatedConspiracy(uint256 id, address creator, string name, string description, string imageUri);
    event Upvoted(address upvoter, uint256 conspiracyId, uint256 latestUpvotes, uint256 latestDownvotes);
    event Downvoted(address downvoter, uint256 conspiracyId, uint256 latestUpvotes, uint256 latestDownvotes);

    constructor(address devWallet, address marketingWallet) ConspiracyDAOToken(devWallet, marketingWallet) {}

    //////////////////////////////////////////
    //////////     Modifiers     ////////////
    ////////////////////////////////////////

    modifier shouldNotBeUpvoted(uint256 conspiracyId){
        require(!upvoted[msg.sender][conspiracyId], "You have already Updvoted this conspiracy");
        _;
    }

    modifier shouldNotBeDownvoted(uint256 conspiracyId) {
        require(!downvoted[msg.sender][conspiracyId], "You have already Downvoted this conspiracy");
        _;
    }

    ///////////////////////////////////////////////////
    //////////     External Functions     ////////////
    /////////////////////////////////////////////////

    function createConspiracy(string calldata name, string calldata description, string calldata imageUri) external {
        require(balanceOf(msg.sender) >= createConspiracyCost, "Your CDAO Token Balance is too low to create a conspiracy!");

        uint256 conspiracyId = conspiracies.length;
        address creator = msg.sender;

        Conspiracy memory conspiracy = Conspiracy({
            name: name,
            id: conspiracyId,
            creator: creator,
            description: description,
            imageUri: imageUri,
            upvotes: 0,
            downvotes: 0
        });

        conspiracies.push(conspiracy);
        emit CreatedConspiracy(conspiracyId, creator, name, description, imageUri);

        _transfer(msg.sender, owner(), createConspiracyCost);
    }

    function upvote(uint256 conspiracyId) external shouldNotBeUpvoted(conspiracyId) {
        address user = msg.sender;
        Conspiracy storage conspiracy = conspiracies[conspiracyId];

        if(downvoted[user][conspiracyId]){
            downvoted[user][conspiracyId] = false;
            conspiracy.downvotes--;
        }
        
        upvoted[user][conspiracyId] = true;
        conspiracy.upvotes++;

        emit Upvoted(user, conspiracyId, conspiracy.upvotes, conspiracy.downvotes);
    }

    function downvote(uint256 conspiracyId) external shouldNotBeDownvoted(conspiracyId){
        address user = msg.sender;
        Conspiracy storage conspiracy = conspiracies[conspiracyId]; 

        if(upvoted[user][conspiracyId]){
            upvoted[user][conspiracyId] = false;
            conspiracy.upvotes--;
        }
        
        downvoted[user][conspiracyId] = true;
        conspiracy.downvotes++;

        emit Downvoted(user, conspiracyId, conspiracy.upvotes, conspiracy.downvotes);
    }

    // Multiplies by 1e18 "So expects a normal number and not a token number"
    function setCreateConspiracyCost(uint256 newCreateConspiracyCost) external onlyOwner {
        createConspiracyCost = newCreateConspiracyCost * 1e18;
    }


    ///////////////////////////////////////////////////
    //////////     External View Functions     ////////////
    /////////////////////////////////////////////////

    function getAllConspiracies() external view returns(Conspiracy[] memory){
        return conspiracies;
    }

    function getConspiracy(uint256 conspiracyId) external view returns(Conspiracy memory){
        return conspiracies[conspiracyId];
    } 

    function getUsersVoteFor(uint256 conspiracyId) external view returns(string memory) {
        address user = msg.sender;
        if(upvoted[user][conspiracyId]){
            return "Upvoted";
        } else if(downvoted[user][conspiracyId]){
            return "Downvoted";
        } else {
            return "Not voted";
        }
    }
}
// File: contracts/ConspiracyDAOToken.sol



//SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "./ERC20Modified.sol";
// import "@openzeppelin/contracts/access/Ownable.sol";

contract ConspiracyDAOToken is ERC20Modified {

    // In percentages
    uint256 public buyTax = 6;
    uint256 public sellTax = 6;
    uint256 public transferTax = 1;

    //////////  Wallets ////////////////////
    address public devWallet;
    address public marketingWallet;

    mapping(address => bool) public isPoolAddress;

    constructor(address _devWallet, address _marketingWallet) ERC20Modified("Conspiracy DAO", "CDAO") {
        _mint(msg.sender, 100000000000e18);
        devWallet = _devWallet;
        marketingWallet = _marketingWallet;
    }   

    //////////////////////////////////////////
    //////////     EVENTS     ///////////////
    ////////////////////////////////////////
    event DeductedBuyTax(address indexed from, uint256 indexed taxPercentage, uint256 indexed taxAmount);
    event DeductedSellTax(address indexed from, uint256 indexed taxPercentage, uint256 indexed taxAmount);
    event DeductedTransferTax(address indexed from, uint256 indexed taxPercentage, uint256 indexed taxAmount);

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     * 
     * Added a Transfer tax of 1% that goes for marketing
     * Added a Buy/Sell tax
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
    */
    function transfer(address to, uint256 amount) public override returns(bool){
        // Sanity Checks
        require(amount > 0, "ERC20: transfer amount cannot be Zero");
        require(to != address(0), "ERC20: transfer to the zero address");
        require(balanceOf(msg.sender) >= amount, "ERC20: transfer amount exceeds balance");

        uint256 newTransferAmount = _deductTax(msg.sender, amount);

        
        _transfer(msg.sender, to, newTransferAmount);
        return true;
    }

    function transferFrom(address from, address to, uint256 amount) public override returns (bool) {
        // Sanity Checks
        require(amount > 0, "ERC20: transfer amount cannot be Zero");
        require(to != address(0), "ERC20: transfer to the zero address");
        require(balanceOf(from) >= amount, "ERC20: transfer amount exceeds balance");

        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        
        uint256 newTransferAmount = _deductTax(from, amount);
        
        _transfer(from, to, newTransferAmount);
        return true;
    }

    function _deductTax(address from, uint256 amount) internal returns(uint256) {
        uint256 taxAmount;

        // Gas Saving
        address _marketingWallet = marketingWallet;
        address _devWallet = devWallet;

        if(isPoolAddress[msg.sender]){
            // Buy/Sell Tax of 6%
            if(from == msg.sender){
                taxAmount = (amount * buyTax)/100;

                uint256 splitTaxAmount = taxAmount/2;
                _balances[_marketingWallet] += splitTaxAmount;
                _balances[_devWallet] += splitTaxAmount;

                emit DeductedBuyTax(from, buyTax, taxAmount);
            } else {
                taxAmount = (amount * sellTax)/100;

                uint256 splitTaxAmount = taxAmount/2;
                _balances[_marketingWallet] += splitTaxAmount;
                _balances[_devWallet] += splitTaxAmount;

                emit DeductedSellTax(from, sellTax, taxAmount);
            }
        } else {
            // Transfer Tax of 1%
            taxAmount = (amount * 1)/100;
            _balances[_marketingWallet] += taxAmount;
            emit DeductedTransferTax(from, transferTax, taxAmount);
        }

        _balances[from] -= taxAmount;
   
        return (amount - taxAmount);
    }

    ///////////////////////////////////////////////////
    //////////    Update Functions        ////////////
    /////////////////////////////////////////////////
    function addPoolAddress(address _poolAddress) external onlyOwner {
        isPoolAddress[_poolAddress] = true;
    }
    function updateDevWallet(address newWalletAddress) external onlyOwner {
        devWallet = newWalletAddress;
    }
    function updateMarketingWallet(address newWalletAddress) external onlyOwner {
        marketingWallet = newWalletAddress;
    }
    function updateTaxes(uint256 newBuyTax, uint256 newSellTax, uint256 newTransferTax) external onlyOwner {
        if(buyTax != newBuyTax) {
            buyTax = newBuyTax;
        }
        if(sellTax != newSellTax) {
            sellTax = newSellTax;
        }
        if(transferTax != newTransferTax) {
            transferTax = newTransferTax;
        }
    }
}
// File: contracts/ERC20Modified.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}


/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20Modified is Context, IERC20, IERC20Metadata, Ownable {
    // Modified _balances to internal
    mapping(address => uint256) internal _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }
}